14 Creep Curve

As has been outlined in Chapter 13 in more detail creep tests are carried out on specimens loaded, e.g., in tension or compression, usually at constant load, inside a furnace which is maintained at a constant temperature. The extension of the specimen is measured as a function of time.

A typical *creep curve* for metals, polymers, and ceramics exsists of three parts and is schematically shown in Fig. 4.1 characterizing the three creep stages called *transient creep, steady creep,* and *accelerating creep.*

In the following MAPLE worksheet an exponential description of a *creep curve* has been represented :

⊙ 14_1.*mws*

Exponential Description

```
> restart:
> epsilon[creep](t):=A[11]*(1-exp(-A[12]*
> sqrt(t)))+A[21]*t+A[31]*(exp(A[32]*t^n)-1);
```

$$\varepsilon_{creep}(t) := A_{11}\,(1 - e^{(-A_{12}\,\sqrt{t})}) + A_{21}\,t + A_{31}\,(e^{(A_{32}\,t^n)} - 1)$$

```
> Digits:=5:
> epsilon[c](t):=subs({A[11]=0.4,A[12]=5,
> A[31]=0.02,A[32]=3, n=10},%%);
```

$$\varepsilon_c(t) := 0.38 - 0.4\,e^{(-5\,\sqrt{t})} + A_{21}\,t + 0.02\,e^{(3\,t^{10})}$$

```
> epsilon[c](0):=evalf(subs(t=0,%));
```

$$\varepsilon_c(0) := 0.$$

```
> epsilon[c](1):=evalf(subs(t=1,%%));
```

$$\varepsilon_c(1) := 0.77902 + A_{21}$$

```
> A[21]:=solve(epsilon[c](1)=1,A[21]);
```

$$A_{21} := 0.22098$$

```
> alias(H=Heaviside,th=thickness,co=color):
> plot1:=plot(epsilon[c](t),t=0..1,th=2):
```

```
> plot2:=plot({epsilon[c](1),epsilon[c](1)
> *H(t-1)}, t=0..1.001):
> plots[display]({plot1,plot2});
```



Fig. 14.1 Creep curve, exponential discription

Time Derivative

```
> time_derivative(t):=diff(epsilon[c](t),t);
```

$$\text{time\_derivative}(t) := \frac{1.0000\, e^{(-5\sqrt{t})}}{\sqrt{t}} + A_{21} + 0.60\, t^9\, e^{(3\, t^{10})}$$

```
> time_derivative(0):=infinity;
```

$$\text{time\_derivative}(0) := \infty$$

```
> time_derivative(1):=evalf(subs({A[21]=0.22098,
> t=1},%%));
```

$$\text{time\_derivative}(1) := 12.280$$

```
> plot3:=plot(time_derivative(t),
> t=0..1,0..2,th=2,co=black):
> plot4:=plot({2,2*H(t-1)},t=0..1.001,co=black):
> plots[display]({plot3,plot4});
```



Fig. 14.2 Time derivative of a creep curve

```
> plots[display]({plot1,plot2,plot3,plot4});
```



Fig. 14.3 Creep curve, exponential discription and time derivative

Creep Parameters

The creep curve exsists of three parts:

```
> restart:
> parameters_of_the_primary_term:=A[11], A[12];
```

$$parameters\_of\_the\_primary\_term := A_{11}, A_{12}$$

```
> parameters_of_the_scondary_term:=A[21]=
> K*sigma^m;    # NORTON-BAILEY
```

$$parameters\_of\_the\_scondary\_term := A_{21} = K\,\sigma^m$$

```
> parameters_of_the_tertiary_term:=A[31],
> A[32],n;
```

$$parameters\_of\_the\_tertiary\_term := A_{31}, A_{32}, n$$

For the *primary creep* the sqrt(t)-law has been assumed, the justification of which has been analysed in Chapter 11.

The exponent n in the tertiary term regulates the tangent of the creep curve at the creep rupture time. The creep parameters can be determined by suitable experiments, some of which are discussed in Chapter 13.

Creep Rate and Acceleration

```
> restart: Digits:=5:
> epsilon[creep](t):=A[11]*(1-exp(-A[12]*
> sqrt(t)))+A[21]*t+A[31]*(exp(A[32]*t^n)-1);
```

$$\varepsilon_{creep}(t) := A_{11}\,(1 - e^{(-A_{12}\sqrt{t})}) + A_{21}\,t + A_{31}\,(e^{(A_{32}\,t^n)} - 1)$$

```
> creep_rate(t):=diff(epsilon[creep](t),t);
```

$$\mathrm{creep\_rate}(t) := \frac{1}{2}\,\frac{A_{11}\,A_{12}\,e^{(-A_{12}\sqrt{t})}}{\sqrt{t}} + A_{21} + \frac{A_{31}\,A_{32}\,t^n\,n\,e^{(A_{32}\,t^n)}}{t}$$

```
> creep_rate(0):=infinity;
```

$$\mathrm{creep\_rate}(0) := \infty$$

```
> creep_rate(1):=subs(t=1,%%);
```

$$\mathrm{creep\_rate}(1) := \frac{1}{2}\,A_{11}\,A_{12}\,e^{(-A_{12})} + A_{21} + A_{31}\,A_{32}\,n\,e^{A_{32}}$$

```
> creep_rate(1):=evalf(subs({A[11]=0.4,A[12]=5,
> A[21]=0.22098,A[31]=0.02,A[32]=3,n=10},%%));
```

$$\mathrm{creep\_rate}(1) := 12.280$$

```
> Creep_rate(t):=evalf(subs({A[11]=0.4,A[12]=5,
> A[21]=0.22098,A[31]=0.02,A[32]=3,n=10},
> creep_rate(t)));
```

$$\mathrm{Creep\_rate}(t) := \frac{1.0000\,e^{(-5.\sqrt{t})}}{\sqrt{t}} + 0.22098 + 0.60\,t^9\,e^{(3.\,t^{10})}$$

```
> acceleration(t):=diff(epsilon[creep](t),t$2);
```

$$\text{acceleration}(t) := -\frac{1}{4}\,\frac{A_{11}\,A_{12}\,e^{(-A_{12}\,\sqrt{t})}}{t^{(3/2)}} - \frac{1}{4}\,\frac{A_{11}\,{A_{12}}^2\,e^{(-A_{12}\,\sqrt{t})}}{t} + \frac{A_{31}\,A_{32}\,t^n\,n^2\,e^{(A_{32}\,t^n)}}{t^2} - \frac{A_{31}\,A_{32}\,t^n\,n\,e^{(A_{32}\,t^n)}}{t^2} + \frac{A_{31}\,{A_{32}}^2\,(t^n)^2\,n^2\,e^{(A_{32}\,t^n)}}{t^2}$$

```
> Acceleration(t):=evalf(subs({A[11]=0.4,A[12]=5,
> A[31]=0.02,A[32]=3,n=10},%),3);
```

$$\text{Acceleration}(t) := -\frac{0.50000\,e^{(-5.\,\sqrt{t})}}{t^{(3/2)}} - \frac{2.5000\,e^{(-5.\,\sqrt{t})}}{t} + 5.40\,t^8\,e^{(3.\,t^{10})} + 18.00\,t^{18}\,e^{(3.\,t^{10})}$$

```
> alias(H=Heaviside, th=thickness, co=color):
> plot1:=plot({Creep_rate(t),Acceleration(t)},
> t=0..1,-10..10,co=black,th=2):
> plot2:=plot({10,-10,10*H(t-1),-10*H(t-1)},
> t=0..1.001,co=black):
> plots[display]({plot1,plot2});
```



Fig. 14.4 Creep rate and acceleration

A similar worksheet has been submitted to the Maplesoft Application Center (www.maplesoft.com) as an Application Demonstration by Betten (2007).

Furthermore, some other worsheets have also been submitted to Maplesoft in several categories, e.g., Mathematics: Engineering-Mathematics, Differential Geometry, Linear Algebra; Engineering: Mechanics, Engineering-Mathematics.